# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=unicode-character-database
pkgver=11.0.0
pkgrel=1
pkgdesc="Unicode Character Database"
arch=(any)
license=(BSD)
url="http://www.unicode.org/"
source=("UCD-$pkgver.zip::http://www.unicode.org/Public/zipped/$pkgver/UCD.zip"
        "Unihan-$pkgver.zip::http://www.unicode.org/Public/zipped/$pkgver/Unihan.zip"
        LICENSE)
noextract=({UCD,Unihan}-$pkgver.zip)
sha512sums=('954a499ff072727c8778e711fb1753834adf09277c9e9a2592c02b59971860f9a407c7e9985ed1e6bdda7a3cf3b1cd6316599c4bffe8b0625413f874486830c1'
            'a045e79545f1ec3529a8556d41c73b20809979e002f98314068c6f91daa79abb2ab3ab383090651b55fb55e343d8fcf67bea822d93d6e3221a206d58ef9e3a7b'
            '9ba78d68ed39745e55538180f132ff47423edb46adde7ac3e62a489870d153808bcf2db57b77e6a7cdae9b4fa03aba99cf228315b9a15be090bedef00b9e69ed')

package() {
  for _f in UCD Unihan; do
    install -Dm644 $_f-$pkgver.zip "$pkgdir/usr/share/unicode/$_f.zip"
    bsdtar -C "$pkgdir/usr/share/unicode" -x --no-same-owner --no-same-permissions -f $_f-$pkgver.zip
  done

  ln -s unicode "$pkgdir/usr/share/$pkgname"

  install -Dm644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
